BY JOHN S. CULLINGS

THE MIGHTY MITE MC-10

Radio Shack's entry into the low-priced micro market reveals its secrets: CoCo power at a poco price.

Meet the MC-10. Radio Shack's new TRS-80 Micro Color Computer is the latest addition to the Color Computer family. At one-sixth the size of a CoCo and at a retail price of $119.95, the new MC-10 will generate much interest with personal-computer enthusiasts (see Photo 1).

What's this computer really like? How compatible is it with the Color Computer? What's under the hood? What hidden features does it have? How are musical tones generated? Is machine-language programming possible? Sit down, relax, and enjoy the potpourri of MC-10 knowledge to follow.

A Quick Example

Let's get started by jumping off at the deep end to answer the last question first. Program Listing 1 is a Basic program that POKEs a machine-language program, a short routine that plays a tune and changes the screen color, into memory. Try it, just for fun. Type in the program exactly as shown (you can omit the remarks at lines 11, 31, and 61).

Type RUN, and press enter. After the OK prompt, type EXEC and press enter. You have just run your first machine-language program on the MC-10! I'll give more information on machine-language programming later in this article. For the curious reader, hex address FFAB is the Basic ROM's entry point for the SOUND command.



Photo 1. The MC-10 Micro Color Computer

A Look Inside

The MC-10 consists of 20 ICs for the computer hardware (Photo 2) and a video modulator (Photo 3). Most of the ICs are from the standard 7400 family. There are four key ICs: the MC6803 microprocessor, MC6847 video-display generator (VDG), and two NEC PD4016, 2048-by-8-bit static RAMs. The MC-10

```
10 REM  MC10 MACHINE LANGUAGE
11 REM  *SOUND & SCREEN DEMO*
15 CLEAR 100,20000
20 FOR  I=20000 TO 20041
25 READ A: POKE I,A: NEXT I
30 POKE 16927,78: POKE 16928,32

31 REM  LDD snd len JSR  FF  AB
35 DATA 204,120,002,189,255,171
40 DATA 204,154,002,189,255,171
45 DATA 204,176,002,189,255,171
50 DATA 204,189,004,189,255,171
55 DATA 204,176,002,189,255,171
60 DATA 204,189,006,189,255,171

61 REM LDAA #88 STA  BF  FF RTS
65 DATA 134,088,183,191,255,057
```

Program Listing 1. Machine-Language Sound and Screen Demo

System Requirements

The MC-10
4K RAM

CoCo Functions Not in MC-10	MC-10 Functions Not in Color Basic
'	LET
ELSE	LPRINT
CSAVE,A	↑
OPEN	SQR
CLOSE	LOG
MOTOR	EXP
AUDIO	COS
SUB	TAN
EOF	VARPTR
JOYSTK	CLOAD*
	CSAVE*

Table 1. Basic Command Differences



Photo 3. The Video Modulator



Photo 2. Inside the MC-10

video modulator is a new design, which uses an MC1372 IC for generating the RF. The old-style modulator used a transistor to generate the RF.

Another function of the new modulator is the generation of the clock signals for the MC6803 microprocessor. The internal clock frequency of the MC6803 is 0.8949 MHz. The MC-10 uses an inexpensive color TV crystal at 3.5795 MHz, which is located inside the video-modulator package. The MC6803 takes the color crystal frequency and divides it by four to get the internal clock frequency of 0.8948 MHZ. The color crystal supplies the color-burst frequency to the MC1372, which generates the colors of the MC-10 computer.

The heart of the MC-10 is the MC6803 microprocessor, which is an 8-bit IC (Fig. 1). The MC6803 is unique in a number of ways. It has 128 bytes of internal RAM; a 16-bit, three function programmable timer; serial interface; and 29 parallel I/O and two handshaking control lines.

One of the outstanding features of the MC6803 is the 29 parallel I/O lines (Fig. 2). Figure 2 shows the MC6803 having four ports. These ports are much the same as a peripheral interface adapter (PIA). A PIA function is to provide a universal means of interfacing peripheral devices to the computer, such as the VDG, printer, or keyboard.

The MC6803's four ports are more than just a PIA. These four ports can function as bidirectional, 8-bit data ports or multiplexed as an address or data bus. Port 1 decodes the keyboard. Port 2 provides interfacing from the cassette and printer. Port 3 is a multiplex port; it is multiplexed between the data bus lines D0 through D7 and the lower-memory address line A0 through A7. Port 4 provides the upper-memory line A8 through A15.

The MC6803 has 128 bytes of internal memory from decimal 128 to 255 or hex 0080 to 00FF. This memory area serves as the operating-system work area. Another feature of the MC6803 is its reduced IC count, made possible by the way the MC6803 handles the 29 I/O lines. This makes the MC6803 a very powerful computer for its low IC count.

The MC-10 memory consists of two NEC PD4016 2K static RAM ICs. (Figure 3 shows its pin configuration.) The RAM goes from decimal 16384 to 20479 or hex 4000 to 4FFF. U9 provides the first 2K of RAM, and U10 provides the second 2K.

Table 4 shows the MC-10's memory map. Decimal 0000 to 0031 is for internal addresses of the ports. Decimal 128 to 255 is the internal RAM of the MC-6803. Decimal 16384 is the start of the video-display area. Basic starts at decimal 17222 and can run up to decimal 20377 for a 4K MC-10. The stack pointers are stored from decimal 20378 to 20479.

The MC-10 uses the MC6847 video-display generator, but not to its full potential. The MC6847 needs 6K of memory for the highest-resolution graphics mode. The MC-10 supports the full color set of the MC6847: four different alphanumeric modes; two semigraphics modes, and nine colors.

CoCo Comparison

How does the MC-10 keyboard compare with the CoCo keyboard? It uses real push-button keys, but the keyboard



Fig. 1. The MC6803 Microprocessor IC



Fig. 2. The MC6803's 29 Parallel I/O Lines

is only three-quarters size. Touch typists won't be pleased. Also, the loss of five keys (48 vs. 53), including the left shift key, makes typing more difficult.

On the positive side, you can enter 38 Basic keywords by pressing the control key and another key together. Also, you can enter all 16 graphic characters with eight different background colors (128 total) by pressing shift and one of 16 alphabetic keys together. The actual screen display is the reverse of the keyboard symbols, however.

Most of the MC-10 Basic commands are the same as the Color Computer (non-Extended) commands. Table 1 summarizes the Basic command differences. The MC-10 command set includes six scientific functions, which require the Extended Basic option on the CoCo. Machine-language programmers take note: The Color Computer Basic commands CLOADM, EXEC, USR, and VARPTR are all included in the MC-10 command set.

Although the Basic commands are similar, most of the Basic machine-code tokens are not. This means that Basic programs on cassette will not run, list or renumber properly on the other type computer.

Fortunately, it seems that cassette tape file formats are compatible. I have been able to SKIPFile, CLOAD, and CLOADM MC-10 cassette programs on my revision E, Extended Color Basic Color Computer. It should be possible to write a translation program to convert MC-10 Basic programs to CoCo Basic programs and vice-versa. Machine-language programs for the MC-10's 6803 microprocessor could be written and assembled on the CoCo. The machine-language cassette tape could then be CLOADMed into the MC-10.

Music Anyone?

The MC-10 operation manual describes the SOUND command, but does not describe which tone number produces which musical note. For example, SOUND 102,14 produces a middle C for one second. Table 2 shows a piano keyboard layout with tone numbers for about 3½ octaves. As the tone number increases, the precision of the musical pitch decreases. Not being particularly musically inclined, I will leave the compositions to you.

9,600 Baud?

For those of you with serial printers that operate at rates of other than 600 baud, don't give up. The MC-10 will run at almost any rate you choose. Although the 6803 microprocessor chip includes a serial output port, Radio Shack has chosen to implement the serial output function the CoCo way. By POKEing a number into location 16931 and 16932 (hex 4223 and 4224), you can select a variety of baud rates. For all baud rates from 300 up, only one POKE is required because location



Table 2. Tone Numbers for Musical Notes

16931 contains a zero for these speeds.

Baud rates from 110–4,800 have worked 100 percent with a Hewlett-Packard Model 85 computer at the other end. A 9,600 baud rate did not work. Apparently the MC-10 skips over the 9,600 baud rate—too fast or too slow. My 9,600-baud buffered printer does work with the MC-10. Try it, maybe it will work for you. Table 3 lists the values to POKE for all the common baud rates.

Teleprinter users might be interested in an end-of-line delay POKE. The two addresses immediately above the baud-rate addresses 16933 and 16934 (hex 4225 and 4226) control the amount of delay at the end of each line. I cannot recommend a precise value, but try POKEing large numbers (less than 255) into address 16933.

Baud	ADDR 16931	ADDR 16932
110	2	163
300	0	241
600	0	118
1200	0	57
2400	0	26
4800	0	10
9600	0	2**

**This baud rate may be too far off for reliable operation.

Table 3. Baud Rates

```
0 REM  MODS TO MIMO

66 A=A+48-7*(A>9)

245 GOSUB30:IF ER=1THEN255
247 POKE SDA,DB:CDB=PEEK(SDA)

320 IF MO=1THEN405
322 GOSUB120

805 PRINT"MIMO COMMANDS:M,D,F,C,J,E"

845 IFC$="E"THENEXEC63278

*** DELETE LINES:
    840,950,955,960,965,970
```

Program Listing 2. Modifications to MIMO Program

```
5 REM  CSAVEM  *USE DECIMAL NRS*
10 POKE 16999,2: GOTO 30
20 A=INT(DA/256):B=DA-256*A:RETURN
30 INPUT"STRT";DA: GOSUB 20
40 POKE 17004,A: POKE 17005,B
50 POKE 17007,A: POKE 17008,B
60 INPUT" END";DA: GOSUB 20
70 POKE 17009,A: POKE 17010,B
80 INPUT"TXFR";DA: GOSUB 20
90 POKE 17002,A: POKE 17003,B
100 INPUT "NAME";N$
110 PRINT"PUSH RECORD AND ENTER"
120 K$=INKEY$:IF K$="" THEN 120
130 EXEC 64603 N$:FOR I=1TO3000
140 NEXT: PRINT"DONE": SOUND66,3
```

Program Listing 3.

MC-10 Monitor

A useful and essential utility for machine-language programming is a monitor which examines and changes individual bytes in memory, and displays large blocks of memory in hex code and in ASCII (alphanumeric) representation.

I decided to modify the MIMO Basic monitor of Sergio Zigras (*80 Micro*, Jan. 1983, pp. 252–256) for the MC-10. Originally written for the 16K RAM Color Computer, this little monitor does an excellent job on the MC-10. It even includes a Find routine to search for any character or string.

(If you do not have a copy of the January *80 Micro*, send a self-addressed, stamped envelope to *HOT CoCo*, Pine St., Peterborough, NH 03458. We will send you a copy of the article.)

The modification of MIMO turned out to be very simple. Only one Basic command is not supported by the MC-10—the ELSE command in lines 66, 245, and 320. Why did Radio Shack eliminate this useful command? Also the Basic reentry address of line 845 needs to be changed to the appropriate MC-10 address (63278).

Finally, I deleted lines 840, 950, 960, 965, and 970 (the Verify function) to save memory. With these changes, a 4K MC-10 has approximately 750 bytes of user memory available for a machine-language program. Use the command CLEAR 100, 19630 to reserve space for such a program. Program Listing 2 lists the necessary changes to the MIMO program.

Pièce de Résistance

Here's where we get to the greatest challenge, and possibly the most interesting facet of the MC-10 Micro CoCo. My goal is not to teach beginning Assembly-language programming, but to point out the real possibility of serious machine-language programming on the little CoCo. MC-10 machine-language programming is quite similar to that of the Color Computer; however, the operation manual gives no clue to this capability.

The 6803 microprocessor is very similar to the 6800, but with some added features such as an internal 128-byte RAM and an internal PIA. It is also similar to the 6809 chip with many iden-

HEX	DECIMAL	FUNCTION
0000-001F	00000-00031	6803 micro I/O
0080-00FF	00128-00255	Direct Page RAM
4000-41FF	16384-16895	Screen Memory
4200-4284	16896-17028	System overhead
4285-42AE	17029-17070	RAM hooks
42B2-4334	17074-17204	Key entry buffer
4346-4F99	17222-20377	BASIC area
4F9A-4FFF	20378-20479	System stack
5000-8FFF	20480-36863	RAM expansion
E000-FFFF	57344-65535	BASIC ROM
FFDC-FFED	65500-65517	ROM subr entry addr
FFF0-FFFF	65520-65535	6803 Interupts

Table 4. MC-10 Memory Map

HEX PNTR ADDR	DECIMAL PNTR ADDR	FUNCTION NAME	FUNCTION DESCRIPTION
FFDC F883	65500 63619	POLCAT	Poll keyboard
FFDE F9C6	65002 63942	CHROUT	Char out
FFE0 FF4E	65504 65358	CRSDON	Tape startup
FFE2 FEB9	65506 65209	BLKIN	Read block
FFE4 FCC0	65508 64704	BLKOUT	Write block
FFE6 FFAB	65510 65451	SNDOUT	Sound out
FFE8 FCB7	65512 64695	WRTLDR	Write leader
FFEA ECE3	65514 60643	GIVABF	2 bytes → BASIC
FFEC EBC7	65516 60359	INTCNV	BASIC → 2 bytes

Table 5. Subroutine Entry Addresses

ical operation codes. However, it does not contain a U or Y register, and it has fewer addressing modes. It is much more difficult to write position-independent code with the 6803.

For more information on the 6803 chip, I recommend reading *6801, 68701, and 6803 Microcomputer Programming and Interfacing*, by Andrew C. Staugaard, Jr. The pocket reference guide by Motorola is also worth having. Ask for the *MC6801/68701/6803 Microcomputer Instruction Set Summary* from Motorola Inc., Integrated Circuits Division, 3501 Ed Bluestein Blvd., Austin, TX 78721.

A necessary Basic command for use with machine-language programming is CSAVEM (save a maching-language program to cassette). This is an Extended Basic command on the CoCo. Program Listing 3 is a short Basic program that performs the CSAVEM function. The start, end, and transfer addresses must be given as decimal numbers, usually near the top of RAM. I use decimal 20000 (hex 4E20) for a convenient starting address.

Efficient machine-language programming on any computer requires a knowledge of its architecture, memory use, ROM subroutine addresses, and RAM hooks. Table 4 shows the overall memory map of the MC-10. Addresses are listed in both decimal and hexadecimal notation.

Of special interest is the table of pointers to useful ROM subroutines. Table 5 lists these subroutine pointers and the addresses to which they point. For information on the operation of these subroutines, refer to Radio Shack's book *Getting Started with Color Basic* section IV, part B, pp. 267–270.

The ROM subroutine entry-address table in the MC-10 is a list of addresses where you find the various subroutines. For example, the POLCAT address of 65500 (hex FFDC) points to 63619 (hex F883). The correct way to use the ROM subroutine entry points is as follows:

```
POLCAT  EQU  65500
        LDX  POLCAT  Get POLCAT
                     pointer
        JSR  0,X     Jump subr there
```

If the X register cannot be used or if memory is to be conserved, POLCAT can be called directly by JSR $F883, but this method is somewhat risky. Some future version of the Basic ROM might not use the same subroutine addresses.

Key-Code Example

Program Listing 4 is a Basic program with an imbedded machine-language subroutine. This program illustrates the use of POLCAT, GIVABF, and USR subroutine calls. This program was adapted from a Color Computer program on page 268 of *Getting Started with Color Basic*.

The machine language subroutine listing (Program Listing 5) is given for reference. You can gain much insight into the differences between 6803 and 6809 machine-language programming by comparing this listing with the Radio Shack original.

The key-code program can be especially useful to the MC-10 user because it shows the ASCII codes for all the graphic characters and keyboard Basic commands. Just type in and run Listing 4. After you press each subsequent key, the character and its ASCII code will display. To get control codes 1–26, press control Z and any character from A to Z. To exit this program press enter.

Inconclusions

The MC-10 keyboard is good. The keyboard is bad. The Basic is powerful—but not powerful enough. The price is low. The price is too high.

Take your pick. A few things are clear to me:

- 4K memory is not enough (more is promised).
- The missing left shift key is a pain in the. . . finger.
- That 6803 is one powerful microprocessor.
- Where's the joystick port???
- The small size is *neat*. Where can I get a matching TV?
- A high-resolution graphics mode is not likely.
- Integrated circuits should be in sockets.
- A direct video output would sure be nice.

Radio Shack, I love/hate you.■

Address correspondence to the author c/o HOT CoCo, *Pine St., Peterborough, NH 03458.*



Fig. 3. Pin Configuration of the NEC PD4016 Static RAM IC

```
10 REM  ASCII KEY CODE
20 CLEAR 100,20000
30 FOR I=20000 TO 20027
40 READ A: POKE I,A: NEXT I
50 DATA 254,255,220,173,000,039
60 DATA 249,129,010,038,010,173
70 DATA 000,039,252,129,065,045
80 DATA 002,128,064,022,079
90 DATA 254,255,234,110,000
100 POKE 16918,78: POKE 16919,32
110 A=USR(0)
120 IF A=13 THEN END
130 PRINT CHR$(A);" =";A
140 GOTO 110
```

Program Listing 4. ASCII Key Code

```
HEX CODE          SOURCE CODE          COMMEMTS

 FE FFDC   LOOP1    LDX   POLCAT   Addr of POLCAT
 AD 00              JSR   0,X      Poll for a key
 27 F9              BEQ   LOOP1    if none, try again
 81 0A              CMPA  #10      Control key
 26 0A              BNE   OUT      Exit if not
 AD 00     LOOP2    JSR   0,X      Yes, get next key
 27 FC              BEQ   LOOP2    if none, try again
 81 41              CMPA  #65      is it A to Z?
 2D 02              BLT   OUT      if lower, exit
 80 40              SUBA  #64      convert to CTRL
 16        OUT      TAB            get rtn byte ready
 4F                 CLRA           zero MSB
 FE FFEA            LDX   GIVABF   Addr of GIVABF
 6E 00              JMP   0,X      go there
           POLCAT   EQU   65500
           GIVABF   EQU   65514
                    END
```

Program Listing 5. The Assembly Version of Listing 4